

# NAG Fortran Library Routine Document

## F08HSF (CHBTRD/ZHBTRD)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F08HSF (CHBTRD/ZHBTRD) reduces a complex Hermitian band matrix to tridiagonal form.

### 2 Specification

```
SUBROUTINE F08HSF(VECT, UPLO, N, KD, AB, LDAB, D, E, Q, LDQ, WORK, INFO)
ENTRY      chbtrd (VECT, UPLO, N, KD, AB, LDAB, D, E, Q, LDQ, WORK, INFO)
INTEGER    N, KD, LDAB, LDQ, INFO
real      D(*), E(*)
complex  AB(LDAB,*), Q(LDQ,*), WORK(*)
CHARACTER*1 VECT, UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

The Hermitian band matrix  $A$  is reduced to real symmetric tridiagonal form  $T$  by a unitary similarity transformation:  $T = Q^H A Q$ . The unitary matrix  $Q$  is determined as a product of Givens rotation matrices, and may be formed explicitly by the routine if required.

The routine uses a vectorisable form of the reduction, due to Kaufman (1984).

### 4 References

Kaufman L (1984) Banded eigenvalue solvers on vector machines *ACM Trans. Math. Software* **10** 73–86

Parlett B N (1980) *The Symmetric Eigenvalue Problem* Prentice-Hall

### 5 Parameters

1: VECT – CHARACTER\*1 *Input*

*On entry:* indicates whether  $Q$  is to be returned as follows:

if VECT = 'V',  $Q$  is returned (and the array  $Q$  must contain a matrix on entry);

if VECT = 'U',  $Q$  is updated (and the array  $Q$  must contain a matrix on entry);

if VECT = 'N',  $Q$  is not required.

*Constraint:* VECT = 'V', 'U' or 'N'.

2: UPLO – CHARACTER\*1 *Input*

*On entry:* indicates whether the upper or lower triangular part of  $A$  is stored as follows:

if UPLO = 'U', then the upper triangular part of  $A$  is stored;

if UPLO = 'L', then the lower triangular part of  $A$  is stored.

*Constraint:* UPLO = 'U' or 'L'.

- 3: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 4: KD – INTEGER *Input*  
*On entry:*  $k$ , the number of super-diagonals of the matrix  $A$  if UPLO = 'U', or the number of sub-diagonals if UPLO = 'L'.  
*Constraint:*  $KD \geq 0$ .
- 5: AB(LDAB,\*) – **complex** array *Input/Output*  
**Note:** the second dimension of the array AB must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  Hermitian band matrix  $A$ , stored in rows 1 to  $k + 1$ . More precisely, if UPLO = 'U', the elements of the upper triangle of  $A$  within the band must be stored with element  $a_{ij}$  in  $AB(k + 1 + i - j, j)$  for  $\max(1, j - k) \leq i \leq j$ ; if UPLO = 'L', the elements of the lower triangle of  $A$  within the band must be stored with element  $a_{ij}$  in  $AB(1 + i - j, j)$  for  $j \leq i \leq \min(n, j + k)$ .  
*On exit:*  $A$  is overwritten.
- 6: LDAB – INTEGER *Input*  
*On entry:* the first dimension of the array AB as declared in the (sub)program from which F08HSF (CHBTRD/ZHBTRD) is called.  
*Constraint:*  $LDAB \geq \max(1, KD + 1)$ .
- 7: D(\*) – **real** array *Output*  
**Note:** the dimension of the array D must be at least  $\max(1, N)$ .  
*On exit:* the diagonal elements of the tridiagonal matrix  $T$ .
- 8: E(\*) – **real** array *Output*  
**Note:** the dimension of the array E must be at least  $\max(1, N - 1)$ .  
*On exit:* the off-diagonal elements of the tridiagonal matrix  $T$ .
- 9: Q(LDQ,\*) – **complex** array *Input/Output*  
**Note:** the second dimension of the array Q must be at least  $\max(1, N)$  if VECT = 'V' or 'U', and at least 1 if VECT = 'N'.  
*On entry:* if VECT = 'U', Q must contain the matrix formed in a previous stage of the reduction (for example, the reduction of a banded Hermitian-definite generalized eigenproblem); otherwise Q need not be set.  
*On exit:* if VECT = 'V' or 'U', the  $n$  by  $n$  matrix  $Q$ .  
Q is not referenced if VECT = 'N'.
- 10: LDQ – INTEGER *Input*  
*On entry:* the first dimension of the array Q as declared in the (sub)program from which F08HSF (CHBTRD/ZHBTRD) is called.  
*Constraints:*  
 $LDQ \geq \max(1, N)$  if VECT = 'V' or 'U';  
 $LDQ \geq 1$  if VECT = 'N'.

- 11: WORK(\*) – **complex** array Workspace  
**Note:** the dimension of the array WORK must be at least max(1, N).
- 12: INFO – INTEGER Output  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed tridiagonal matrix  $T$  is exactly similar to a nearby matrix  $A + E$ , where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$  is a modestly increasing function of  $n$ , and  $\epsilon$  is the **machine precision**.

The elements of  $T$  themselves may be sensitive to small perturbations in  $A$  or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

The computed matrix  $Q$  differs from an exactly unitary matrix by a matrix  $E$  such that

$$\|E\|_2 = O(\epsilon),$$

where  $\epsilon$  is the **machine precision**.

## 8 Further Comments

The total number of real floating-point operations is approximately  $20n^2k$  if VECT = 'N' with  $10n^3(k-1)/k$  additional operations if VECT = 'V'.

The real analogue of this routine is F08HEF (SSBTRD/DSBTRD).

## 9 Example

To compute all the eigenvalues and eigenvectors of the matrix  $A$ , where

$$A = \begin{pmatrix} -3.13 + 0.00i & 1.94 - 2.10i & -3.40 + 0.25i & 0.00 + 0.00i \\ 1.94 + 2.10i & -1.91 + 0.00i & -0.82 - 0.89i & -0.67 + 0.34i \\ -3.40 - 0.25i & -0.82 + 0.89i & -2.87 + 0.00i & -2.10 - 0.16i \\ 0.00 + 0.00i & -0.67 - 0.34i & -2.10 + 0.16i & 0.50 + 0.00i \end{pmatrix}.$$

Here  $A$  is Hermitian and is treated as a band matrix. The program first calls F08HSF (CHBTRD/ZHBTRD) to reduce  $A$  to tridiagonal form  $T$ , and to form the unitary matrix  $Q$ ; the results are then passed to F08JSF (CSTEQR/ZSTEQR) which computes the eigenvalues and eigenvectors of  $A$ .

### 9.1 Program Text

**Note:** the listing of the example program presented below uses **bold italicised** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08HSF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
      INTEGER                NIN, NOUT
      PARAMETER              (NIN=5, NOUT=6)
      INTEGER                NMAX, KMAX, LDAB, LDQ
      PARAMETER              (NMAX=8, KMAX=8, LDAB=KMAX+1, LDQ=NMAX)
```

```

*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, KD, N
      CHARACTER        UPLO
*      .. Local Arrays ..
      complex          AB(LDAB,NMAX), Q(LDQ,NMAX), WORK(NMAX)
      real              D(NMAX), E(NMAX-1), RWORK(2*NMAX-2)
      CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
      EXTERNAL         X04DBF, chbtrd, csteqr
*      .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08HSF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX .AND. KD.LE.KMAX) THEN
*
*          Read A from data file
*
          READ (NIN,*) UPLO
          IF (UPLO.EQ.'U') THEN
              DO 20 I = 1, N
                  READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
20             CONTINUE
          ELSE IF (UPLO.EQ.'L') THEN
              DO 40 I = 1, N
                  READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
40             CONTINUE
          END IF
*
*          Reduce A to tridiagonal form T = (Q**H)*A*Q (and form Q)
*
          CALL chbtrd('V',UPLO,N,KD,AB,LDAB,D,E,Q,LDQ,WORK,INFO)
*
*          Calculate all the eigenvalues and eigenvectors of A
*
          CALL csteqr('V',N,D,E,Q,LDQ,RWORK,INFO)
*
          WRITE (NOUT,*)
          IF (INFO.GT.0) THEN
              WRITE (NOUT,*) 'Failure to converge.'
          ELSE
*
*              Print eigenvalues and eigenvectors
*
              WRITE (NOUT,*) 'Eigenvalues'
              WRITE (NOUT,99999) (D(I),I=1,N)
              WRITE (NOUT,*)
              IFAIL = 0
*
              CALL X04DBF('General',' ',N,N,Q,LDQ,'Bracketed','F7.4',
+                  'Eigenvectors','Integer',RLABS,'Integer',CLABS,
+                  80,0,IFAIL)
*
          END IF
          END IF
          STOP
*
99999 FORMAT (8X,4(F7.4,11X,:))
      END

```

## 9.2 Program Data

F08HSF Example Program Data

```

4 2                                :Values of N and KD
'L'                               :Value of UPLO
(-3.13, 0.00) ( 1.94,-2.10) (-3.40, 0.25)
( 1.94, 2.10) (-1.91, 0.00) (-0.82,-0.89) (-0.67, 0.34)
(-3.40,-0.25) (-0.82, 0.89) (-2.87, 0.00) (-2.10,-0.16)
                (-0.67,-0.34) (-2.10, 0.16) ( 0.50, 0.00) :End of matrix A

```

## 9.3 Program Results

F08HSF Example Program Results

Eigenvalues  
           -7.0042                  -4.0038                  0.5968                  3.0012

Eigenvectors

	1	2	3	4
1	( 0.7293, 0.0000)	(-0.2128, 0.1511)	(-0.3354,-0.1604)	(-0.5114,-0.0163)
2	(-0.1654,-0.2046)	( 0.7316, 0.0000)	(-0.2804,-0.3413)	(-0.2374,-0.3796)
3	( 0.6081, 0.0301)	( 0.3910,-0.3843)	(-0.0144, 0.1532)	( 0.5523, 0.0000)
4	( 0.1653,-0.0303)	( 0.2775,-0.1378)	( 0.8019, 0.0000)	(-0.4517, 0.1693)

---